



DANTE Sample A&B Programming

Sample Timing and Data Collection

Sample Programming

Sample programs are text scripts specifying how DANTE controls modules and the COM to collect and record data. DANTE has two sample programs, A and B. The sample programs are run on a simple daily schedule set with the SCHEDULE command.

The SCHEDULE Command

SCHEDULE A OFFSET PERIOD

SCHEDULE B OFFSET PERIOD

The SCHEDULE command tells DANTE to start running either program A or B at OFFSET minutes after midnight and run the program again every PERIOD minutes until the end of the day.

OFFSET –minutes from midnight (00:00) when DANTE will run the program for the first time.

PERIOD – period in minutes at which to run program until the end of the day.

Basic Sample Program Structure

Sample Programs are little more than a list of commands representing a non-branching and non-looping algorithm¹. Each line of a program represents a single command. When a command is completed the next command is processed. This repeats until the end of the file is reached.

Handling Time

DANTE's clock requires external power. If external power is disconnected then DANTE's clock reverts to 2007. Prior to running a sample program DANTE checks its clock. If the time is not valid DANTE automatically updates its clock from COM unit (which is powered by the communications backup battery).

Sample Scheduling

Program A overrides B

```
SCHEDULE A 0 60
SCHEDULE B 0 10
00:00 Run A (ignore B)
00:10 Run B
00:20 Run B
...
00:50 Run B
01:00 Run A (ignore B)
```

If a program's run time exceeds its time slot then samples are skipped

```
SCHEDULE A 0 60
SCHEDULE B 0 10
Assume program A takes
15 minutes to run
00:00 Run A
00:10 A still running, so
DANTE does not start
program B
00:15 A finishes
00:20 Run B
```

¹ Branching requires conditional statements. Looping requires statements such as GOTO, FOR or WHILE. These statements are not currently supported, but may be added in the future.



DANTE automatically sets the time in all modules before running a sample program. This allows driver scripts running in modules to use the `TIMESTAMP` command to record the current time in the data stream.

We strongly recommend all users use GPS time everywhere in the DANTE system. Use the following commands in your programs to keep all times in the DANTE system set to GPS time with +/- 2 second accuracy.

```
COM ENABLE
COMTIME
COM SEND "GPS ON\r"
... do other things for 45 seconds while GPS gets a fix ...
COM SEND "SETGPSTIME\r"
```

DANTE Sample Program A/B Command List

COMTIME

Sets the DANTE clock from the COM unit clock.

DELAY

Waits a specified number of milliseconds.

```
DELAY 1000
```

Waits one second.

START

Tells a module to start running its driver script. This command accepts either a Module ID (MID) or Local ID (LID).

```
START 001
```

Tells module 001 to run its driver script.

SLEEP

Tells a module to enter low power sleep mode. Use this command only when the sample program is done talking to a module. This command accepts either a Module ID (MID) or Local ID (LID). This command overrides any script in process, disabling the module's serial port and disconnecting switched power outputs unless they are set to PERM. It is not possible to wake a module from sleep mode within a sample program. Modules wake automatically each time a sample program is run.

```
SLEEP 001
```

Tells module 001 to go to sleep.

WAITFOR

Waits for a module's driver script to end. This command accepts either a Module ID (MID) or Local ID (LID). The command times out after three minutes if the module's script does not end, or at any time if the module fails to respond.

```
WAITFOR 001
```

Waits for module 001's driver script to terminate.

**GETDATA**

Retrieves data from a module and stores it in the sample data stream. This command accepts either a Module ID (MID) or Local ID (LID).

```
GETDATA 001
```

Retrieves data from module 001.

COMPOWER ON / OFF

Enables or disables the 12V power output to the COM unit. The COM power output charges the COM battery. It is not required to communicate with the COM unit.

COM ENABLE

Wakes the COM unit from sleep and enables COM power (allowing the COM to charge its battery).

COM DISABLE

Switches the COM unit back to sleep mode and removes COM power, terminating COM battery charging.

COM SEND

Sends a command to the COM unit. Enables the COM unit first if it is not already enabled. A CR is automatically sent to the COM after the command. An optional timeout parameter is allowed (in milliseconds). If no timeout is provided the default is five seconds. The minimum timeout is 50 milliseconds and the maximum is two minutes (120000 milliseconds). Invalid timeout parameters are ignored.

```
COM SEND "gps on" 1000
```

Sends the command GPS ON followed by CR to the COM unit. No record is written to sample data if COM accepts the command. If the COM does not respond within the 1000 millisecond timeout period or responds with an error then either

```
<Error line='10'>com send timeout</Error>
```

or

```
<Error line='10'>com send error</Error>
```

is written in the sample data stream.

COM LOG

Sends a command to the COM unit and records the response to the sample data stream wrapped in an xml tag. Enables the COM first if it is not already enabled. A CR is automatically sent to the COM after the command. Up to five seconds is allowed for the COM to respond. If the COM does not respond or responds with an error then either

```
<Error line='10'>com log timeout</Error>
```

or

```
<Error line='10'>com log error</Error>
```

is written in the sample data stream before the COM response.

```
COM LOG GPS_FIX "gps fix"
```

Sends 'gps fix' to the COM followed by CR. The response is recorded in the sample data stream like this:

```
<GPS_FIX>
```

```
GPS_FIX
```

```
GPSF 47 37.767 N, 122 14.788 W, T=033033, D=051113
```

```
OK - 0 Events
```

```
</GPS_FIX>
```

**ADC RESET**

Resets all averaging data in DANTE's analog to digital converter system.

ADC AVERAGE

Records the average ADC values into the sample data stream. The values recorded are the average since the last ADC RESET or since DANTE woke from sleep mode.

ADC MIN

Records the minimum ADC values into the sample data stream. The values recorded are the minimum since the last ADC RESET or since DANTE woke from sleep mode.

ADC MAX

Records the maximum ADC values into the sample data stream. The values recorded are the maximum since the last ADC RESET or since DANTE woke from sleep mode.

ADC LAST

Records the most recent ADC values into the sample data stream (with no averaging).

ADC Raw Data Commands

ADC RAVERAGE, ADC RMIN, ADC RMAX and ADC RLAST are the same as their equivalents above, but record raw ADC data not converted to engineering units.

GETEC

Records DANTE's event counters to the sample data stream and resets the event counters. This should always be included near the end of a sample program, and may be used multiple times in a program to help identify the source of any events.

MODPOWER OFF

This command has no immediate effect – it tells DANTE to disable the 12V and VB (5V) power supplies from modules at the end of the sample program, disabling all module switched power outputs between programs. The setting persists until changed manually or in another sample program.

NOTE: MODPOWER ON is not a valid command in sample programs because DANTE always switches on module power before sample programs are run (if power was not already enabled).

MODPOWER PERM

This command has no immediate effect – it tells DANTE to leave the 12V and VB (5V) power connected to modules at the end of the sample program, allowing modules to keep switched power outputs enabled between programs. The setting persists until changed manually or in another sample program.

COMTYPE

Changes the COMTYPE setting in DANTE's configuration.

PARSERTYPE

Changes the PARSERTYPE setting in DANTE's configuration.

SAMPLESPERTRANSMIT

Changes the SAMPLESPERTRANSMIT setting in DANTE's configuration.



Common Sample Program Code

Most sample programs include common elements to capture event counters from both DANTE and the COM unit and measure battery voltage. Many programs also activate the COM's GPS receiver and ACM (tilt & compass sensor). Please refer to the COM unit Command Line Interface documentation for information about commands sent to the COM unit.

NOTE: DANTE does not currently support comments in sample programs!

Start of Sample Program

COM ENABLE	Wake the COM unit. This allows the com battery to charge
COM SEND "acm start"	Start the Accelerometer & Compass
COM SEND "gps on"	Power the GPS receiver
COM LOG COMEC "getec"	Read the event counters from the COM and records them in
COM SEND "resetec"	Reset the COM's event counters
START 001	Start module 001's driver script
START 002	Start module 002's driver script
SLEEP 003	Put any installed but unused modules to sleep to save power
SLEEP 004	Modules 001, 002, 003 and 004 are always installed.

Data Collection from Modules and the COM Unit

DELAY 5000	DELAY saves some power if we know module 001's driver script takes at least 5 seconds to run
WAITFOR 001	Waits up to three minutes for module 001's driver to finish
GETDATA 001	Retrieves and stores data from module 001
SLEEP 001	Put module to sleep when done – this saves power.
WAITFOR 002	Same process for module 002
GETDATA 002	
SLEEP 002	
COM SEND "gps wff" 91000	Wait for a gps position fix in the COM unit, up to 91 seconds.
COM LOG GPS_FIX "gps fix"	Get the gps fix
COM SEND "gps off"	Turn off the GPS receiver
COM LOG ACM_LOG "acm average"	Read the average from the COM's Accelerometer & Compass
COM SEND "acm stop"	Stop the COM's Accelerometer & Compass
COM SEND "sleep"	Put the COM unit to sleep when not needed to save power



Ending the Sample Program

Sample programs usually end by recording DANTE's event counters (a powerful verification and debugging tool) and some logging of Analog to Digital Converter (ADC) results.

COM DISABLE	This also occurs automatically when the sample program ends.
GETEC	Stores DANTE's event counters in the sample data stream and resets the counters
ADC AVERAGE	Average ADC values allow power estimation
ADC MAX	Maximum ADC values are nice to know.